
Natural Source  

0010 * Module:   ZZMATCH                    (c) 2002 LEN Consulting LLC 
0020 * System:   ZZUtils 
0030 * Function: Sequential Match 
0040 * Author:   R. Zbrog 
0050 * Date:     02/18/89 
0060 ************************************************************************ 
0070 *   Date   *   Name    * Summary of Changes 
0080 ************************************************************************ 
0090 * 09/11/01 * R. Zbrog  * Demo files 
0100 * 09/23/94 * R. Zbrog  * Nat 2.2.5 
0110 * 09/29/93 * R. Zbrog  * Comments requesting custom code. 
0120 * 12/16/92 * R. Zbrog  * Trace 
0130 ************************************************************************ 
0140 * 
0150 * Sample sequential-match program. 
0160 * 
0170 *   Matches an Adabas file and a work file  -  the Adabas file MUST be 
0180 *   processed as the outer loop.  Two sequential files may be matched by 
0190 *   replacing the Adabas READ loop with a READ WORK loop (although no 
0200 *   starting value may be specified for the READ WORK loop).  Two Adabas 
0210 *   files may not be matched; one of them must be converted to a work 
0220 *   file (ie the inner loop MUST process a WORK file). 
0230 * 
0240 *   The view or record layout for the outer loop is called MSTR.  The 
0250 *   record layout for the inner loop is called TRAN. 
0260 * 
0270 *   The program uses the standard algorithm for a master file updated by 
0280 *   a transaction file.  The view or record layout for the outer loop 
0290 *   is called MSTR; the record layout for the inner loop is called TRAN. 
0300 *   The program detects record matches, extra MSTR records, and extra 
0310 *   TRAN records. 
0320 * 
0330 *   The MSTR file must have a unique key and the TRAN file must be in 
0340 *   the same sequence as MSTR.  If both files have unique keys, a ratio 
0350 *   of 1:1, then either file may be defined as MSTR or TRAN.  If there 
0360 *   is a 1:n ratio, then the file with the unique key must be defined as 
0370 *   MSTR and the file with the non-unique key must be defined as TRAN. 
0380 * 
0390 *   Indeed, the files may not be true master and transaction files.  It 
0400 *   is the one-to-many ratio that defines which file will be treated as 
0410 *   master and which will be treated as transaction. 
0420 * 
0430 *   If MSTR is an Adabas file, the program may be a good Prefetch 
0440 *   candidate. 
0450 * 
0460 *   When customizing this program, review all comments for coding 
0470 *   suggestions.  For example, if a 1:1 ratio, optimize for this. 
0480 * 
0490 *   Lines marked with '<<<' contain mandatory changes. 
0500 *   Lines marked with '>>>' contain optional code; use as appropriate. 
0510 *   Remove optional code if it is not needed, as this will improve 
0520 *   performance. 
0530 * 
0540 *   It is recommended that the #TRACE-ALL variable be set to TRUE until 
0550 *   the user has reached an acceptable level of confidence in the 
0560 *   program's use.  This setting will cause an audit report to be 
0570 *   produced, displaying all widows, orphans, and matches. 

0580 * 
0590 *   Software AG's EMPLOYEES demonstration file was used as the basis for 
0600 *   the sample match in this program.  To ensure unique keys, the MSTR 
0610 *   file is defined as a WORK file.  Refer to module ZZMATCHJ for JCL 
0620 *   which includes sample MSTR and TRAN data. 
0630 * 
0640 * 
0650 DEFINE DATA LOCAL 
0660 1 MSTR (A72)       1 REDEFINE MSTR     /* Extract of EMPLOYEES demo file 
0670   2 PERSONNEL-ID (A8) 
0680   2 NAME (A20) 
0690   2 FIRST-NAME (A20) 
0700   2 PHONE (A15) 
0710   2 DEPT (A6) 
0720   2 LANG (A3) 
0730 * 
0740 * 
0750 1 #TRAN (A40)      1 REDEFINE #TRAN 
0760   2 NAME (A20) 
0770   2 DEPT (A6) 
0780   2 CUSTOMER (N5) 
0790   2 COMMISSION (N7.2) 
0800 * 
0810 * 
0820 1 #START                               /* >>> Optional FROM processing 
0830   2 DEPT (A6)                INIT <' '> 
0840   2 NAMEW (A20)              INIT <' '> 
0850                         1 REDEFINE #START 
0860   2 #START-KEY (A26) 
0870 * 
0880 1 #END                                 /* >>> Optional THRU processing 
0890   2 DEPT (A6)                INIT <'99'> 
0900   2 NAME (A20)               INIT <'99999999'> 
0910                         1 REDEFINE #END 
0920   2 #END-KEY(A26) 
0930 * 
0940 1 #MKEY                      /* >>> Build key from non-contiguous fields 
0950   2 DEPT (A6) 
0960   2 NAME (A20) 
0970                         1 REDEFINE #MKEY 
0980   2 #MSTR-KEY (A26) 
0990 * 
1000 1 #TKEY                      /* >>> Build key from non-contiguous fields 
1010   2 DEPT (A6) 
1020   2 NAME (A20) 
1030                         1 REDEFINE #TKEY 
1040   2 #TRAN-KEY (A26) 
1050 * 
1060 * 
1070 1 #N                                   /* Counters 
1080   2 CNTM (P7)                          /* Master 
1090   2 EXTM (P7)                          /* Extra Master (Widow) 
1100   2 CNTT (P7)                          /* Transaction 
1110   2 EXTT (P7)                          /* Extra Trans (Orphan) 
1120   2 MCHT (P7)                          /* Matched transactions 
1130 * 
1140 1 #LOGICALS 
1150   2 #MSTR-MATCHED (L) 
1160   2 #TRAN-NEEDED (L)         INIT <TRUE> 

1170   2 #EOF (L)                 INIT <FALSE> 
1180   /* 
1190   2 #TRACE-ALL (L)           INIT <TRUE>    /* >>> 
1200   2 #TRACE-MSTR (L)          INIT <FALSE>   /* >>> 
1210   2 #TRACE-TRAN (L)          INIT <FALSE>   /* >>> 
1220   2 #TRACE-MISC (L)          INIT <FALSE>   /* >>> 
1230 * 
1240 1 #MISC 
1250   2 #DATE (A8)               INIT <*DATU>   /* Constant for report 
1260   2 #TIME (A5)               INIT <*TIMX>   /* Constant for report 
1270 * 
1280   2 #NAME (A20) 
1290 END-DEFINE 
1300 * 
1310 FORMAT (01) PS=60 LS=132 
1320 */ 
1330 * 
1340 READ WORK 1 MSTR                       /* >>> READ MSTR BY DEPT-PERSON 
1350   /*      FROM #START-KEY              /* >>> 
1360   ASSIGN #MKEY.DEPT = MSTR.DEPT        /* >>> Build a 
1370   ASSIGN #MKEY.NAME = MSTR.NAME        /* >>> contiguous key 
1380   IF  #MSTR-KEY > #END-KEY             /* >>> Optional 'thru' 
1390     THEN 
1400       ESCAPE BOTTOM 
1410   END-IF 
1420   ADD 1 TO #N.CNTM 
1430   ASSIGN #MSTR-MATCHED = FALSE         /* Not yet matched 
1440 * 
1450 * 
1460   COMPRESS MSTR.NAME                   /* <<< 
1470            ',' 
1480       INTO #NAME LEAVING NO SPACE 
1490   COMPRESS #NAME                       /* <<< 
1500            MSTR.FIRST-NAME 
1510       INTO #NAME 
1520 * 
1530 * 
1540   RPT. 
1550   REPEAT 
1560     IF  #TRAN-NEEDED 
1570       THEN 
1580         TR. 
1590         READ WORK 2 ONCE #TRAN 
1600              AT END OF FILE 
1610                ASSIGN #EOF = TRUE 
1620              END-ENDFILE 
1630         IF  #EOF 
1640           THEN 
1650             MOVE ALL H'FF' TO #TRAN-KEY 
1660             ASSIGN #TRAN-NEEDED = FALSE 
1670           ELSE 
1680             ASSIGN #TKEY.DEPT = #TRAN.DEPT  /* >>> Build a 
1690             ASSIGN #TKEY.NAME = #TRAN.NAME  /* >>> contiguous key 
1700             ADD 1 TO #N.CNTT 
1710         END-IF 
1720     END-IF 
1730 * 
1740     DECIDE FOR FIRST CONDITION 
1750       WHEN #MSTR-KEY < #TRAN-KEY       /* Must get next master 

1760            IF  NOT #MSTR-MATCHED       /* Previously matched 
1770              THEN                      /* Process an unmatched master 
1780                IF  #TRACE-ALL          /* >>> 
1790                 OR #TRACE-MSTR 
1800                  THEN 
1810                    WRITE (01) 
1820                               MSTR.DEPT 
1830                            2X #NAME 
1840                            2X MSTR.PHONE 
1850                            2X MSTR.PERSONNEL-ID 
1860                            2X MSTR.LANG 
1870                           88T #N.CNTM (EM=Z,ZZZ,ZZ9) 
1880                          110T 'Extra Master'          /* Widow 
1890                END-IF 
1900                /* 
1910                /* <<<  Code 'widow' logic here.  <<< 
1920                /* 
1930                ADD 1 TO #N.EXTM 
1940            END-IF 
1950            ASSIGN #TRAN-NEEDED = FALSE /* Apply to next master 
1960            ESCAPE BOTTOM (RPT.)        /* Process next master 
1970       WHEN #MSTR-KEY = #TRAN-KEY       /* Process a match 
1980            IF  #TRACE-ALL              /* >>> 
1990             OR #TRACE-MSTR 
2000             OR #TRACE-TRAN 
2010              THEN 
2020                WRITE (01) 
2030                           MSTR.DEPT 
2040                        2X #NAME 
2050                        2X MSTR.PHONE 
2060                        2X MSTR.PERSONNEL-ID 
2070                        2X MSTR.LANG 
2080                       64T #TRAN.CUSTOMER 
2090                       74T #TRAN.COMMISSION (EM=Z,ZZZ,ZZ9.99) 
2100                       88T #N.CNTM (EM=Z,ZZZ,ZZ9) 
2110                       99T #N.CNTT (EM=Z,ZZZ,ZZ9) 
2120                      110T 'Matched MSTR/TRAN' 
2130            END-IF 
2140            ADD 1 TO #N.MCHT 
2150            /* 
2160            /* <<<  Code 'match' logic here.  <<< 
2170            /* 
2180            ASSIGN #MSTR-MATCHED = TRUE 
2190            ASSIGN #TRAN-NEEDED  = TRUE /* Try next transaction 
2200 * >>>      ESCAPE BOTTOM (RPT.)        /* Next master; force 1:1 ratio 
2210       WHEN #MSTR-KEY > #TRAN-KEY       /* Process an unmatched tran 
2220            IF  #TRACE-ALL              /* >>> 
2230             OR #TRACE-TRAN 
2240              THEN 
2250                WRITE (01) 
2260                           #TRAN.DEPT 
2270                        2X #TRAN.NAME 
2280                       64T #TRAN.CUSTOMER 
2290                       74T #TRAN.COMMISSION (EM=Z,ZZZ,ZZ9.99) 
2300                       99T #N.CNTT (EM=Z,ZZZ,ZZ9) 
2310                      110T 'Extra Transaction'    /* Orphan 
2320            END-IF 
2330            ADD 1 TO #N.EXTT 
2340            PERFORM ORPHAN 

2350            ASSIGN #TRAN-NEEDED = TRUE  /* Try next transaction 
2360       WHEN NONE 
2370            IGNORE 
2380     END-DECIDE 
2390   END-REPEAT 
2400 END-WORK                               /* >>> END-READ 
2410 * 
2420 IF  #TRAN-NEEDED                       /* No waiting transaction 
2430  OR #EOF                               /* Transaction end of file 
2440   THEN 
2450     IGNORE 
2460   ELSE 
2470     IF  #TRACE-ALL                     /* >>> 
2480      OR #TRACE-TRAN 
2490       THEN 
2500         WRITE (01) 
2510                    #TRAN.DEPT 
2520                 2X #TRAN.NAME 
2530                64T #TRAN.CUSTOMER 
2540                74T #TRAN.COMMISSION (EM=Z,ZZZ,ZZ9.99) 
2550                99T #N.CNTT (EM=Z,ZZZ,ZZ9) 
2560               110T 'Trailing TRAN'     /* Orphan 
2570     END-IF 
2580     ADD 1 TO #N.EXTT 
2590     PERFORM ORPHAN 
2600 END-IF 
2610 * 
2620 REPEAT WHILE NOT #EOF                  /* Process extra TRAN records 
2630   READ WORK 2 ONCE #TRAN               /* Same file number as (TR.) 
2640        AT END OF FILE 
2650          ESCAPE BOTTOM 
2660        END-ENDFILE 
2670   ADD 1 TO #N.CNTT 
2680   IF  #TRACE-ALL                       /* >>> 
2690    OR #TRACE-TRAN 
2700     THEN 
2710       WRITE (01) 
2720                  #TRAN.DEPT 
2730               2X #TRAN.NAME 
2740              64T #TRAN.CUSTOMER 
2750              74T #TRAN.COMMISSION (EM=Z,ZZZ,ZZ9.99) 
2760              99T #N.CNTT (EM=Z,ZZZ,ZZ9) 
2770             110T 'Extra TRAN > Last MSTR'   /* Orphan 
2780   END-IF 
2790   ADD 1 TO #N.EXTT 
2800   PERFORM ORPHAN 
2810 END-REPEAT 
2820 * 
2830 * 
2840 NEWPAGE (01) 
2850 WRITE (01) 
2860    //     #N.CNTM (EM=Z,ZZZ,ZZ9) 'Master Records Read' 
2870    //     #N.CNTT (EM=Z,ZZZ,ZZ9) 'Transaction Records Read' 
2880  ////     #N.EXTM (EM=Z,ZZZ,ZZ9) 'Unmatched Masters (Widows)' 
2890    //     #N.EXTT (EM=Z,ZZZ,ZZ9) 'Unmatched Transactions (Orphans)' 
2900  ////     #N.MCHT (EM=Z,ZZZ,ZZ9) 'Matched Transactions' 
2910 * 
2920 * 
2930 WRITE (01) TITLE LEFT 

2940            *program (cd=bl) 
2950        10T 'LEN Consulting LLC' 
2960       118T #DATE 
2970            #TIME 
2980     /  10T 'Natural Utilities' 
2990       122t 'ZZUtils v2' (yei) 
3000     /  10T 'Sample Sequential Match Program' (bli) 
3010       120T 'Page:' *PAGE-NUMBER (01) (EM=ZZ,ZZ9) 
3020    //      'Dept'                                /* <<< 
3030         9T 'Name'                                /* <<< 
3040        31T 'Phone'                               /* <<< 
3050        48T 'Personnel'                           /* <<< 
3060        58T 'Lang'                                /* <<< 
3070        64T '  Cust'                              /* <<< 
3080        74T '  Commission'                        /* <<< 
3090        88T '   Master'                           /* <<< 
3100        99T '    Trans'                           /* <<< 
3110       110T 'Message'                             /* <<< 
3120     / 
3130 * 
3140 * 
3150 DEFINE SUBROUTINE ORPHAN 
3160 * 
3170 * <<<  Code 'orphan' logic here.  <<< 
3180 * 
3190 ESCAPE ROUTINE                     /* <<< Replace with 'orphan' logic 
3200 END-SUBROUTINE 
3210 END 

MVS JCL  

0010 //jobname  JOB (acct),'ZZMATCH SAMPLE',CLASS=a,MSGCLASS=x, 
0020 //             MSGLEVEL=(1,1),NOTIFY=userid,REGION=0M 
0030 //*                                    (c) 2002 LEN Consulting LLC 
0040 //MATCH    EXEC NATURAL 
0050 //CMPRINT   DD SYSOUT=* 
0060 //CMWKF01   DD *                       MSTR 
0070 00000050BOSTON              FRANK               202 555 1212   ACCTNGENG 
0080 00000060SMITH               JOHN                714 555 9876   ENTMNTGER 
0090 00000070SMYTHE              JON                 562 555 6543   ENTMNTGER 
0100 00000080GERSTNER            LOU                 204 555 2345   SALE01FRE 
0110 00000090SMITH               SAM                 415 555 6382   SALE01ENG 
0120 00000001CAIN                ABLE                800 555 1234   TECH01ENG 
0130 /* 
0140 //CMWKF02   DD *                       TRAN 
0150 SMITH               ENTMNT07290000025000 
0160 SMYTHE              ENTMNT00284000059900 
0170 GERSTNER            SALE0100185000025000 
0180 GERSTNER            SALE0102974100000000 
0190 GERSTNER            SALE0103297001500000 
0200 GERSTNER            SALE0104126800750000 
0210 SMITH               SALE0100039701000000 
0220 CAIN                TECH0101343000100000 
0230 SMITH               TECH0127032000085500 
0240 THOMPSON            TECH0100719000785000 
0250 THOMPSON            TECH0124900000456700 
0260 ZILDJAN             TECH1300020000012500 
0270 /* 
0280 //CMSYNIN   DD * 
0290 LOGON ZZUTIL 
0300 ZZMATCH 
0310 FIN 
0320 /* 
0330 // 

Sample Report  

ZZMATCH  LEN Consulting LLC                                                                                          09/12/01 12:48 
         Natural Utilities                                                                                               ZZUtils v2 
         Sample Sequential Match Program                                                                               Page:      1  

Dept    Name                  Phone            Personnel Lang    Cust      Commission     Master      Trans  Message  

ACCTNG  BOSTON, FRANK         202 555 1212     00000050  ENG                                   1             Extra Master 
ENTMNT  SMITH, JOHN           714 555 9876     00000060  GER     7290          250.00          2          1  Matched MSTR/TRAN 
ENTMNT  SMYTHE, JON           562 555 6543     00000070  GER      284          599.00          3          2  Matched MSTR/TRAN 
SALE01  GERSTNER, LOU         204 555 2345     00000080  FRE      185          250.00          4          3  Matched MSTR/TRAN 
SALE01  GERSTNER, LOU         204 555 2345     00000080  FRE     2974    1,000,000.00          4          4  Matched MSTR/TRAN 
SALE01  GERSTNER, LOU         204 555 2345     00000080  FRE     3297       15,000.00          4          5  Matched MSTR/TRAN 
SALE01  GERSTNER, LOU         204 555 2345     00000080  FRE     4126    8,007,500.00          4          6  Matched MSTR/TRAN 
SALE01  SMITH, SAM            415 555 6382     00000090  ENG       39    7,010,000.00          5          7  Matched MSTR/TRAN 
TECH01  CAIN, ABLE            800 555 1234     00000001  ENG     1343        1,000.00          6          8  Matched MSTR/TRAN 
TECH01  SMITH                                                   27032          855.00                     9  Trailing TRAN 
TECH01  THOMPSON                                                  719        7,850.00                    10  Extra TRAN > Last MSTR 
TECH01  THOMPSON                                                24900        4,567.00                    11  Extra TRAN > Last MSTR 
TECH13  ZILDJAN                                                    20          125.00                    12  Extra TRAN > Last MSTR    

ZZMATCH  LEN Consulting LLC                                                                                          09/12/01 12:48 
         Natural Utilities                                                                                               ZZUtils v2 
         Sample Sequential Match Program                                                                               Page:      2  

Dept    Name                  Phone            Personnel Lang    Cust      Commission     Master      Trans  Message    

        6 Master Records Read  

       12 Transaction Records Read    

        1 Unmatched Masters (Widows)  

        4 Unmatched Transactions (Orphans)    

        8 Matched Transactions 


